- **We check that the Tamm-Rubilar tensor density defined in kappaLib coincides with the Tamm-Rubilar tensor density in**

  Hehl, Obukhov: Foundations of classical electromagnetics, 2003.

In[1]:=
```
SetDirectory["~/writing/WIP/KappaLib/"];
<< KappaLib.m
```

```
KappaLib v1.1
```

- **Define a general (2,2)-tensor**

In[3]:=
```
kappa = emGeneralKappa["k"];
emKappaToMatrix[kappa] // MatrixForm
```

Out[4]//MatrixForm=

$$\begin{pmatrix} k11 & k12 & k13 & k14 & k15 & k16 \\ k21 & k22 & k23 & k24 & k25 & k26 \\ k31 & k32 & k33 & k34 & k35 & k36 \\ k41 & k42 & k43 & k44 & k45 & k46 \\ k51 & k52 & k53 & k54 & k55 & k56 \\ k61 & k62 & k63 & k64 & k65 & k66 \end{pmatrix}$$

- **Step 1: Compute Fresnel equation for kappa using kappaLib**

In[5]:=
```
xi = {xi0, xi1, xi2, xi3};
generalFresnel = emKappaToFresnel[kappa, xi];
```

- **Step 2: Compute Fresnel equation for kappa using equations D2.45 -- D2.49 in HO:2003.**

- **Convenience functions**

In[7]:=
```
{AA, BB, CC, DD} = emKappaToABCD[kappa];
Am[i_, j_] := AA[[i]][[j]]
Bm[i_, j_] := BB[[i]][[j]]
Cm[i_, j_] := CC[[i]][[j]]
Dm[i_, j_] := DD[[i]][[j]]

Ctrace = Sum[Cm[i, i], {i, 1, 3}];
Dtrace = Sum[Dm[i, i], {i, 1, 3}];
```

- **Define M**

In[14]:=
```
M0 = Det[AA];
```

- **Define M^a**

In[15]:=
```
M1[a_] := -Sum[
    Signature[{b, c, d}] (Am[b, a] Am[c, e] Cm[d, e] + Am[a, b] Am[e, c] Dm[e, d]),
    {b, 1, 3}, {c, 1, 3}, {d, 1, 3}, {e, 1, 3}]
```

- **Define M^ab**

In[16]:=
```
preM2term1[a_, b_] := 1 / 2 Am[a, b] (
    Ctrace^2 + Dtrace^2 -
     Sum[(Cm[c, d] + Dm[d, c]) (Cm[d, c] + Dm[c, d]), {c, 1, 3}, {d, 1, 3}])

preM2term2[a_, b_] :=
 Sum[(Cm[d, c] + Dm[c, d]) (Am[c, a] Cm[b, d] + Dm[d, a] Am[b, c]), {c, 1, 3}, {d, 1, 3}]

preM2terms345[a_, b_] :=
 -Ctrace Sum[Am[c, a] Cm[b, c], {c, 1, 3}] - Sum[Dm[c, a] Am[b, c], {c, 1, 3}] Dtrace -
  Sum[Am[d, c] Cm[a, c] Dm[d, b], {c, 1, 3}, {d, 1, 3}]

preM2term6[a_, b_] := Sum[
   (Am[a, b] Am[d, c] - Am[d, a] Am[b, c]) Bm[d, c],
   {c, 1, 3},
   {d, 1, 3}]

preM2[a_, b_] :=
 preM2term1[a, b] + preM2term2[a, b] + preM2terms345[a, b] + preM2term6[a, b]

M2[a_, b_] := (preM2[a, b] + preM2[b, a]) / 2
```

Define M^abc

```
In[22]:= M3pre[a_, b_, c_] :=
          Sum[Signature[{d, e, c}] (Bm[d, f] (Am[a, b] Dm[e, f] - Dm[e, a] Am[b, f]) +
              Bm[f, d] (Am[a, b] Cm[f, e] - Am[f, a] Cm[b, e]) +
              Cm[a, f] Dm[e, b] Dm[d, f] + Dm[f, a] Cm[b, e] Cm[f, d]),
           {d, 1, 3},
           {e, 1, 3},
           {f, 1, 3}]

In[23]:= M3[a_, b_, c_] := Module[{res, loop, perms},

           perms = Permutations[{a, b, c}];
           res = 0;

           For[loop = 1, loop ≤ Length[perms], loop++,
            res += M3pre[perms[[loop]][[1]], perms[[loop]][[2]], perms[[loop]][[3]]];
           ];

           res / Length[perms]
          ]
```

Define M^abcd

```
In[24]:= M4pre[a_, b_, c_, d_] := Sum[
           Signature[{e, f, c}] Signature[{g, h, d}]
            Bm[h, f] (1 / 2 Am[a, b] Bm[g, e] - Cm[a, e] Dm[g, b]),
           {e, 1, 3},
           {f, 1, 3},
           {g, 1, 3},
           {h, 1, 3}
          ]

In[25]:= M4[a_, b_, c_, d_] := Module[{res, loop, perms},

           perms = Permutations[{a, b, c, d}];
           res = 0;

           For[loop = 1, loop ≤ Length[perms], loop++,
            res += M4pre[perms[[loop]][[1]],
                perms[[loop]][[2]], perms[[loop]][[3]], perms[[loop]][[4]]];
           ];

           res / Length[perms]
          ]
```

Compute Fresnel equation using equation D.2.44

```
In[26]:= qx = {xi1, xi2, xi3};

In[27]:= decompFresnel = xi0^4 M0 + xi0^3 Sum[qx[[i]] M1[i], {i, 1, 3}] +
            xi0^2 Sum[qx[[i]] qx[[j]] M2[i, j], {i, 1, 3}, {j, 1, 3}] +
            xi0 Sum[qx[[i]] qx[[j]] qx[[k]] M3[i, j, k], {i, 1, 3}, {j, 1, 3}, {k, 1, 3}] +
            Sum[qx[[i]] qx[[j]] qx[[k]] qx[[l]] M4[i, j, k, l],
             {i, 1, 3}, {j, 1, 3}, {k, 1, 3}, {l, 1, 3}];
```

Check that the two expressions for the Fresnel equation match

```
In[28]:= Expand[generalFresnel - decompFresnel]

Out[28]= 0
```

Yes, they coincide.